

!----- 1B -----

! 1. Make a simple FODO cell of L cell = 100 m. Each quad is
! L quad = 5 m long. Put the start of the first quadrupole at the
! start of the sequence. Each quad has a focal length of
! f = 200 m ($K1 \times L \text{ quad} = 1/f$ in thin lens approximation).

! 2. Define a proton beam at E tot = 2 GeV. Activate the
! sequence, try to find the periodic solution and plot the
! β -functions. If you found $\beta \text{ max} \approx 460$ m you succeeded.

! 3. Using the plot you obtained can you estimate the phase
! advance of the cell? Compare with the tunes obtained
! from the TWISS.

! --- $\sin(\mu/2) = (b_{\text{max}} - b_{\text{min}})/(b_{\text{max}} + b_{\text{min}})$
! --- $\cos(\mu) = 1 - L^2 / 2f^2$ (L is half cell here)

! 4. Try with E tot = 0.7 GeV: what is the MADX error message?
! --- The beam energy needs to be at least the rest mass

! 5. Try with f = 20 m: what is the MADX error message?
! --- The lattice is now unstable

////////////////////////////////////

f = 200;

Lq = 5;

QF: QUADRUPOLE, L=Lq, K1=1/f/Lq;

QD: QUADRUPOLE, L=Lq, K1=-1/f/Lq;

JUAS: SEQUENCE, REFER=entry, L=100;

qf1: QF, at=0;

qd1: QD, at=50;

ENDSEQUENCE;

beam, particle=proton, energy=2;

!beam, particle=proton, energy=0.7;


use, sequence=JUAS;

twiss, file="juas.twi";

plot, HAXIS=s, VAXIS=betx,bety, colour=100, interpolate;

quit;

////////////////////////////////////



!----- 2A -----

! Consider the FODO cell of tutorial 1 (L cell = 100 m,
! L quad = 5 m and f = 200 m).
! I Define the beam (proton at E tot = 7 TeV), activate the
! sequence and try to twiss it powering the quads to obtain
! $\Delta\mu \approx 90$ deg phase advance in the cell using the thin lens
! approximation (use Fig. 1). What is the actual phase
! advance computed by MADX?

! --- from figure: $K1 L Lq \approx 2.8 \rightarrow K1 = 0.0056$
! --- $\text{mux}_{\text{mad}} = 0.236 \text{ } 2\pi \text{ rad} = 84.96^\circ$

////////////////////////////////////

```
f = 200;  
Lq = 5;  
QF: QUADRUPOLE, L=Lq, K1=0.0056;  
QD: QUADRUPOLE, L=Lq, K1=-0.0056;  
  
JUAS: SEQUENCE, REFER=entry, L=100;  
qf1: QF, at=0;  
qd1: QD, at=50;  
ENDSEQUENCE;
```


```
beam, particle=proton, energy=7000;
```

```
use, sequence=JUAS;
```

```
twiss, file="juas.twi";  
plot, HAXIS=s, VAXIS=betx,bety, colour=100, interpolate;
```

```
quit;
```

////////////////////////////////////



!----- 2B -----

! What is the β max ? Compare with the thin lens approximation (Fig. 2). Compute the maximum beam σ assuming $\epsilon_n = 3$ mrad mm, $E_{\text{tot}} = 7$ TeV?

! --- $\beta_{x_max} = 160.6$ m
 ! --- $\sigma = \sqrt{\epsilon_n/g * \beta} = \sqrt{3e-6/7460 * 160} = 0.25$ mm

! Halve the focusing strength of the quadrupoles, what is the effect of it on the β max , β min and on the $\Delta\mu$? Compare with the parametric plots in Fig. 1 and Fig. 2.

!----- 3A -----

! Consider now that in the cell of Tutorial 2 there are 4 sector
! dipoles of 10 m (assume 5 m of drift space between
! magnets). In the ring there are a total of 736 dipoles with
! equal bending angles. Install the four dipoles in the FODO
! cell. Do the dipoles (weak focusing) affect on the β max and
! the dispersion? Compute the relative variation on the β max
! on the two planes.

! --- betx_max = 160.5m -> it was 160.6 m before. y stays the same
! --- rel = (160.6-160.5)/160.6 = 0.06 %

! From the phase advance of the FODO cell compute the
! horizontal and vertical tune of the machine?

! --- phase adv of a cell = 0.24
! --- total ph adv = 0.24 * 736/4 = 44.16

////////////////////////////////////

f = 200;
Lq = 5;
Ld = 10;
nBend=736;

QF: QUADRUPOLE, L=Lq, K1=0.0056;
QD: QUADRUPOLE, L=Lq, K1=-0.0056;
BM: SBEND, L=Ld, angle=2*pi/nBend;

JUAS: SEQUENCE, REFER=entry, L=100;
qf: QF, at=0;
b1: BM, at=10;
b2: BM, at=25;
qd: QD, at=50;
b3: BM, at=60;
b4: BM, at=75;
ENDSEQUENCE;

beam, particle=proton, energy=7000;

use, sequence=JUAS;

MATCH, SEQUENCE=juas;
GLOBAL, Q1=60.2/8/23; //H-tune
GLOBAL, Q2=67.2/8/23; //V-tune
VARY, NAME= qf.K1, STEP=0.00001;
VARY, NAME= qd.K1, STEP=0.00001;
LMDIF, CALLS=50, TOLERANCE=1e-6;//method adopted
ENDMATCH;

```
twiss, file="juas.twi";
plot, HAXIS=s, VAXIS=betx,bety,dx,dy, colour=100, interpolate;

quit;
////////////////////////////////////
```


!----- 3B -----

! Change the beam to E tot = 3.5 TeV. What is the new tune of the machine? Why?

! --- new ph adv = 0.24 = previous (magnets are scaled)

! Suppose you want to set a tune of (60.2, 67.2), match the FODO to get it. What is the maximum tune that you can reach with 23 cells/octant and 8 octants? (HINT: what is the maximum phase advance per FODO cell in thin approximation?...)

! --- 180° per fodo cell * 23*8 cells.



!----- 4A -----

! Build a transfer line of 10 m with 4 quads of $L=0.4$ m
! (centered at 2, 4, 6, and 8 m). With $K1$ respectively of 0.1,
! 0.1 , 0.1 , 0.1 m^{-2} . Can you find a periodic solution?

! --- no, the lattice is not stable in either planes

! Can you find a Initial Condition solution starting from
! $(\beta x , \alpha x , \beta y , \alpha y) = (1, 0, 2, 0)$?

! --- Sure!

! What is the final optical condition $(\beta x \text{ end } , \alpha x \text{ end } , \beta y \text{ end } , \alpha y \text{ end })$?

! --- x: 50.05527223 m -2.04007934
! --- y: 98.13871917 m -13.56203145

!----- 4B -----

! Starting from $(\beta x , \alpha x , \beta y , \alpha y) = (1 \text{ m}, 0, 2 \text{ m}, 0)$ match the
! line to $(\beta x , \alpha x , \beta y , \alpha y) = (2, 0, 1, 0)$ at the end.

! Starting from $(\beta x , \alpha x , \beta y , \alpha y) = (1 \text{ m}, 0, 2 \text{ m}, 0)$ and the
! gradient obtained with the previous matching, match to
! $(\beta x \text{ end } , \alpha x \text{ end } , \beta y \text{ end } , \alpha y \text{ end })$. Can you find back $K1$
! respectively of
! 0.1, 0.1 , 0.1 , 0.1 $1/m^2$?

! --- it converges to a different solution

! Consider that the quadrupoles have an excitation current
! factor of 100 Am^2 and an excitation magnetic factor of 100
! T/m/A and aperture of 40 mm diameter. Compute the
! magnetic field at the poles of the four quads after matching
! (HINT: assume linear regime and use a dimensional
! approach).

! --- Current = $100 \text{ Am}^2 * K1 = 10 \text{ A}$ (with $K=0.1$)
! --- Field = $100 \text{ T/m/A} * \text{Current} * \text{Aperture} = 20 \text{ T}$ (with 10 A)

////////////////////////////////////

```
Kq = 0.1;
Lq = 0.4;
```

```
QUAD: QUADRUPOLE, L=Lq, K1=Kq;
```

```
JUAS: SEQUENCE, REFER=centre, L=10;
Q1: QUAD, at=2;
Q2: QUAD, at=4;
Q3: QUAD, at=6;
Q4: QUAD, at=8;
ENDSEQUENCE;
```

```
beam, particle=proton, energy=7000;
```


```
use, sequence=JUAS;
```

```
MATCH, SEQUENCE=JUAS, betx=1, bety=2;
constraint, betx=50.055, range=#e;
constraint, alfx=-2.0401, range=#e;
constraint, bety=98.139, range=#e;
constraint, alfy=-13.5620, range=#e;
VARY, NAME= kq1, STEP=0.00001;
VARY, NAME= kq2, STEP=0.00001;
VARY, NAME= kq3, STEP=0.00001;
VARY, NAME= kq4, STEP=0.00001;
LMDIF, CALLS=500, TOLERANCE=1e-6;//method adopted
ENDMATCH;
```

```
twiss, betx=1, alfx=0, bety=2, alfy=0, file="juas.twi";
plot, HAXIS=s, VAXIS=betx,bety,colour=100, interpolate;
```

```
quit;
```

```
////////////////////////////////////
```



!----- 5A-----!

Chromaticity and sextupoles

- ! 1. After the definition of the sequence, convert it in thin lenses with the commands:
! MAKETHIN, SEQUENCE = MY_SEQUENCE;
! use, sequence = MY_SEQUENCE;
! This step is required to allow particle tracking in MAD-X.
- ! 2. With a matching block adjust the tunes of the cell to 0.25.
- ! 3. Using the chromaticities obtained from the twiss,
! compute the tunes for $\Delta p/p = 10e-3$.

! --- $\Delta Q = dq1 * \Delta p/p = -0.318*8*23 * 1e-3 = -0.059$
- ! 4. Track a particle with initial coordinates
! x, y, px, py = (1, 1, 0, 0) mm in 100 cells. Plot the x-px phase space (use gnuplot). How does the particle move in the phase space, cell after cell? Do you see the tunes?
- ! --- The particle always comes back to the same four points.
- ! 5. Track a particle with initial coordinates
! x, y, px, py = (100, 100, 0, 0) mm in 100 cells. Plot the x-px phase space. Does something change with respect to the previous case? Why?
- ! --- Same plot as before, but with larger amplitude.
! --- OK: it's a linear machine!
- ! 6. Repeat point 4 adding DELTAP = 0.01 to the track command. How does the phase space look now? Is the tune still the same? It may help to look only at the first few (4) turns, to get a clearer picture.
- ! --- The particle has now a smaller tune and slips behind.

!----- 5B -----

Non-linearities and large amplitude oscillations.

- ! 7. Add 0.5 m long sextupoles attached to the two quadrupoles. With a matching block adjust the vertical and horizontal chromaticity of the cell (global parameters dq1, dq2) to zero, by powering the two sextupoles (K2_1 and K2_2).
- ! 8. using the obtained K2 1 and K2 2 , β -function and dispersion at sextupoles location, evaluate using the formulas the sextupolar effect on the Q1 for a particle at DELTAP = 1e-3. Compare the results with the value obtained in point 1.

```

! ---  $\Delta Q = \text{Sum over the sexts of } (\beta * K2L * D * \Delta p/p)$ 
!       =  $8*23/4/\pi*(147.5 * 0.0165 * 2.417 * 1e-3 -$ 
!          $34.44 * 0.0301 * 1.173 * 1e-3) = 0.068$ 
! --- The discrepancy comes from the different off-momentum beta

! 9. Repeat point 4 adding DELTAP = 0.01 to the track
! command. Did you manage to recover the original tune
! for the off-momentum particle?

! --- Yep, the tune is now close to 0.25 again

! 10. Repeat point 5. What is going on now?

! --- With a larger initial amplitude we are losing the particle
! --- We are outside the dynamic aperture

! 11. Move the tunes to (0.23, 0.23) and repeat the previous
! point. Is now the particle stable?

! --- The particle looks more stable than before. Moving away
! --- from the resonance we improved the dynamic aperture

```

```

////////////////////////////////////

```

```

!! General parameters

```

```

Lcell = 100;

```

```

nBend=736;

```

```

!! Dipole Parameters

```

```

Ld = 15;

```

```

Ad = 2*pi/nBend;

```

```

!! Quadrupole Parameters

```

```

K1F = 5e-3;

```

```

K1D = -K1F;

```

```

Lq = 5;

```

```

!! Sextupole Parameters

```

```

K2F = 0.0;

```

```

K2D = -K2F;

```

```

Ls = 0.5;

```

```

QF: QUADRUPOLE, L:=Lq, K1:=K1F;

```

```

QD: QUADRUPOLE, L:=Lq, K1:=K1D;

```

```

BM: SBEND, L=Ld, angle:=Ad;

```

```

SF: SEXTUPOLE, L:=Ls, K2:=K2F;

```

```

SD: SEXTUPOLE, L:=Ls, K2:=K2D;

```

```

JUAS: SEQUENCE, REFER=entry, L=100;

```

```

qf: QF, at=0;

```

```
S1 : SF, at=5;
b1:  BM, at=10;
b2:  BM, at=30;
qd:  QD, at=50;
S2 : SD, at=55;
b3:  BM, at=60;
b4:  BM, at=80;
ENDSEQUENCE;
```

```
beam, particle=proton, energy=7000;
```

```
MAKETHIN, SEQUENCE=JUAS;
use, sequence=JUAS;
```

```
//*****//
!   MATCHING OF THE TUNES
//*****//
```

```
match, sequence=JUAS;
!! Variables
vary,name=K1F,step=0.0001;
vary,name=K1D,step=0.0001;
!! Constraints
global, Q1=0.25;
global, Q2=0.25;
!! The next line ask MAD-X to do the matching itself
LMDIF, calls = 1000, tolerance=1E-12;
endmatch;
```

```
//*****//
!   MATCHING OF THE CHROMATICITY
//*****//
```

```
!match, sequence=JUAS;
!!! Variables
!vary,name=K2F,step=0.0001;
!vary,name=K2D,step=0.0001;
!!! Constraints
!global, dq1=0.0;
!global, dq2=0.0;
!!! The next line ask MAD-X to do the matching itself
!LMDIF, calls = 1000, tolerance=1E-12;
!endmatch;
```

```
//*****//
!   TWISS
//*****//
```

```
SELECT,FLAG=TWISS, column=name,s,betx,bety,dx,K1L,K2L;
twiss,file="juas.twi";
```


```
plot, HAXIS=s, VAXIS=betx,bety,dx,dy,colour=100, interpolate;
```

```
//*****  
! TRACKING  
//*****
```

```
track,dump, DELTAP=0.01;  
start, x= 1e-3, px=0, y= 1e-3, py=0;  
start, x= 1e-1, px=0, y= 1e-1, py=0;  
run,turns=4;  
endtrack;
```

```
plot, file="MAD_track",table=track,haxis=x,vaxis=px,  
particle=1,2, colour=100;  
plot, file="MAD_track",table=track,haxis=y,vaxis=py,  
particle=1,2, colour=100;
```

```
quit;  
////////////////////////////////////
```



!----- 6 -----

! Retrieve the LHC injection optics from the repository.
! Download the LHC Run 1 protons, injection optics from
! <http://lhc-optics.web.cern.ch/lhc-optics/www/>
!
! Build a the MADX scripts to call the file and to twiss the
! machine.
! What is the LHC length? What is the s-position of IP1 and
! IP5? and the β -functions there?
! What are the beam1 and beam2 tunes at injections?
! Are the two beams colliding in IP1 at injection?

! Retrieve the collision optics.
! Is the crossing of the two beams vertical or horizontal in
! IP1 at collision?
! What are the beta function at the IPs at collision energy?
! Why do we inject with a higher β -function at the IPs?

! --- For instance:
! `wget http://lhc-optics.web.cern.ch/lhc-optics/www/opt2015/inj/lhc_opt2015_inj.seq`
! `wget http://lhc-optics.web.cern.ch/lhc-optics/www/opt2015/coll400/lhc_opt2015_coll400.seq`

////////////////////////////////////
Option, -echo,warn,-info;

call, file="lhc_opt2015_inj.seq";
!call, file="lhc_opt2015_coll400.seq";

use, sequence=lhcb1;
!use, sequence=lhcb2;

SELECT,FLAG=TWISS, column=keyword,name,s,betx,bety,x,y,dx,dy;
twiss, file="lhc.twi";
plot, HAXIS=s, VAXIS=betx,bety,dx,dy,colour=100;
plot, HAXIS=s, VAXIS=x,y,colour=100;
////////////////////////////////////

! --- then look in the plots and in the Twiss table

